[bookmark: _GoBack]10000 '"SQRSUM/BAS
10100 CLS
10200 CLEAR 600
10300 DEFSTR B
10400 DEFINT C - Y
10500 RR = &F800
10600 IF PEEK(3) = 23 THEN RR = &3C00:IF PEEK(4) = 64 THEN CMD "V80"
10700 B0 = CHR$(017) + CHR$(066) + CHR$(0) + CHR$(025) + CHR$(229) + CHR$(058) +
       CHR$(003) + CHR$(0) + CHR$(254) + CHR$(023) + CHR$(062) + CHR$(004) + CHR
      $(196) + CHR$(094) + CHR$(008) + CHR$(204) + CHR$(213) + CHR$(068) + CHR$(
      225) + CHR$(243)
10800 B1 = CHR$(024) + CHR$(048) + CHR$(017) + CHR$(045) + CHR$(0) + CHR$(025) +
       CHR$(229) + CHR$(058) + CHR$(003) + CHR$(0) + CHR$(254) + CHR$(023) + CHR
      $(196) + CHR$(009) + CHR$(027) + CHR$(204) + CHR$(009) + CHR$(009) + CHR$(
      227) + CHR$(017)
10900 B2 = CHR$(024) + CHR$(0) + CHR$(235) + CHR$(025) + CHR$(193) + CHR$(113) +
       CHR$(035) + CHR$(112) + CHR$(058) + CHR$(003) + CHR$(0) + CHR$(254) + CHR
      $(023) + CHR$(062) + CHR$(004) + CHR$(194) + CHR$(113) + CHR$(008) + CHR$(
      195) + CHR$(210)
11000 B3 = CHR$(068) + CHR$(0) + CHR$(049) + CHR$(050) + CHR$(051) + CHR$(052) +
       CHR$(010) + CHR$(043) + CHR$(053) + CHR$(192) + CHR$(054) + CHR$(010) + C
      HR$(001) + CHR$(251) + CHR$(255) + CHR$(009) + CHR$(229) + CHR$(221) + CHR
      $(225) + CHR$(058)
11100 B4 = CHR$(003) + CHR$(0) + CHR$(254) + CHR$(023) + CHR$(032) + CHR$(004) +
       CHR$(058) + CHR$(090) + CHR$(0) + CHR$(245) + CHR$(017) + CHR$(0) + CHR$(
      0) + CHR$(001) + CHR$(032) + CHR$(004) + CHR$(126) + CHR$(035) + CHR$(007)
       + CHR$(056)
11200 B5 = CHR$(056) + CHR$(032) + CHR$(028) + CHR$(235) + CHR$(229) + CHR$(205)
       + CHR$(017) + CHR$(066) + CHR$(126) + CHR$(113) + CHR$(018) + CHR$(225) +
       CHR$(035) + CHR$(019) + CHR$(016) + CHR$(244) + CHR$(005) + CHR$(221) + C
      HR$(112) + CHR$(0)
11300 B6 = CHR$(241) + CHR$(050) + CHR$(090) + CHR$(0) + CHR$(211) + CHR$(132) +
       CHR$(042) + CHR$(072) + CHR$(064) + CHR$(024) + CHR$(022) + CHR$(229) + C
      HR$(205) + CHR$(245) + CHR$(008) + CHR$(225) + CHR$(119) + CHR$(121) + CHR
      $(229) + CHR$(205)
11400 B7 = CHR$(250) + CHR$(008) + CHR$(225) + CHR$(035) + CHR$(016) + CHR$(241)
       + CHR$(005) + CHR$(221) + CHR$(112) + CHR$(0) + CHR$(042) + CHR$(083) + C
      HR$(018) + CHR$(030) + CHR$(001) + CHR$(123) + CHR$(233) + CHR$(032) + CHR
      $(022) + CHR$(235)
11500 B8 = CHR$(229) + CHR$(205) + CHR$(017) + CHR$(066) + CHR$(026) + CHR$(119)
       + CHR$(225) + CHR$(035) + CHR$(019) + CHR$(016) + CHR$(245) + CHR$(221) +
       CHR$(112) + CHR$(0) + CHR$(241) + CHR$(050) + CHR$(090) + CHR$(0) + CHR$(
      211) + CHR$(132)
11600 B9 = CHR$(201) + CHR$(126) + CHR$(229) + CHR$(205) + CHR$(250) + CHR$(008)
       + CHR$(225) + CHR$(035) + CHR$(016) + CHR$(247) + CHR$(221) + CHR$(112) +
       CHR$(0) + CHR$(201)
11700 BB = B0 + B1 + B2 + B3 + B4 + B5 + B6 + B7 + B8 + B9
11800 LABEL "Length"
11900 B = "0123456789"
12000 N = 0
12100 L = 0
12200 ON STOP GOTO 0
12300 INPUT !0, 0, #1, 95, USING B, "Number of squares per side (3-9) ";N
12400 IF N < 3 OR N > 9 THEN "Length"
12500 DIM H(N), V(N), M(N * N), I(N * N)
12600 LABEL "Draw"
12700 CLS
12800 S = 2 * INT(67 / N)
12900 BU = "####"
13000 A = INT(9 / N)
13100 XL = S * N + 8
13200 'Horizontal lines
13300 FOR X = 10 TO XL - 1
13400 FOR Y = 3 TO A * 6 * N + 3 STEP A * 6
13500 SET(X, Y)
13600 NEXT
13700 NEXT
13800 'Vertical lines
13900 FOR X = 8 TO XL STEP S
14000 FOR Y = 3 TO A * 6 * N + 3
14100 SET(X, Y)
14200 SET(X + 1, Y)
14300 NEXT
14400 NEXT
14500 'Rows
14600 FOR Y = 1 TO N
14700 PRINT !Y * S / 2 + INT(4 - S / 4) - 2, 0, USING BU;Y;
14800 NEXT
14900 'Cols
15000 FOR X = 1 TO N
15100 PRINT !0, X * 2 * A - A + 1, USING "##";X;
15200 NEXT
15300 IF Q = 11 THEN GOSUB "MATRIX"
15400 LABEL "Column"
15500 P = 1
15600 DI = 1
15700 PRINT !0, 22, CHR$(030);
15800 LINE INPUT !0, 22, %Q, LEN(STR$(N)) - 1, 95, USING MID$(B, 4 - LEN(STR$(N)
      ), N), "Column (X) ";BX
15900 IF Q = 11 THEN "Draw"
16000 IF Q = 2 THEN "Done"
16100 X = VAL(BX)
16200 IF X > N OR X = 0 THEN "Column"
16300 U = POS(0) + LEN(BX) - LEN(STR$(N)) + 1
16400 LABEL "Row"
16500 PRINT !U, 22, CHR$(030);
16600 LINE INPUT !U, 22, %Q, LEN(STR$(N)) - 1, 95, USING MID$(B, 4 - LEN(STR$(N)
      ), N), ", Row (Y) ";BY
16700 IF Q = 8 THEN "Column"
16800 IF Q = 2 THEN "Done"
16900 Y = VAL(BY)
17000 IF Y > N OR Y = 0 THEN "Row"
17100 E = POS(0) + LEN(BY) - LEN(STR$(N)) + 1
17200 GOSUB "MSPOT"
17300 LABEL "Over"
17400 Q = FRE(B)
17500 DEF USR = WPEEK(VARPTR(BB) + 1) + LEN(B0) + 2
17600 ON STOP GOTO "No beep"
17700 Q = USR (RR + X * S / 2 + INT(4 - S / 4) - 2 + INT(Y * A * 2 - A + 1) * 80
      )
17800 LABEL "Value"
17900 PRINT !E, 22, ", Value ";STRING$(4, 95);
18000 INPUT !E + 8, 22, %Q, 1, 95, USING B + "-", BV
18100 IF Q = 13 THEN PRINT !E, 22, ", Value ";STRING$(4, 32);:GOTO "No beep"
18200 IF Q = 8 OR Q = 2 THEN "No beep"'if 8 back to row
18300 IF Q > 0 AND Q < 32 THEN "Value"
18400 BM = ""
18500 LINE INPUT !E + 9, 22, %Q, 3, 95, USING B, BM
18600 IF Q = 8 THEN "Value"
18700 IF Q <> 13 THEN IF Q > 2 AND Q < 32 THEN "Value"
18800 IF BV = "-" THEN M((X - 1) * N + Y) = -VAL(BM) ELSE M((X - 1) * N + Y) = V
      AL(BV) * 10 [ LEN(BM) + VAL(BM)
18900 LABEL "No beep"
19000 MN = LEN(B0) + LEN(B1) + LEN(B2) + LEN(B3) + LEN(B4)
19100 MID$(BB, MN, 1) = CHR$(024)
19200 CALL WPEEK(VARPTR(BB) + 1)
19300 MID$(BB, MN, 1) = CHR$(056)
19400 ON STOP GOTO 0
19500 IF Q = 8 AND K => 0 THEN "Row"
19600 LABEL "Done"
19700 IF Q = 2 THEN P = 2
19800 PRINT !0, 23, CHR$(030);
19900 ON P GOSUB "Partial Total", "USER"
20000 FOR K = 1 TO N - 1
20100 IF H(K) <> H(K + 1) OR V(K) <> V(K + 1) THEN EXIT "Column"
20200 NEXT
20300 IF LL <> LR OR LL <> H(K) THEN "Column"
20400 IF P = 2 THEN "Good"
20500 PRINT !0, 22, "Sorting... " CHR$(030);
20600 SORT N * N, * I(1), M(1)
20700 PRINT "Checking ..."
20800 FOR U = 1 TO N * N - 1
20900 IF M(I(U)) = M(I(U + 1)) THEN "Fail"
21000 NEXT
21100 LABEL "Good"
21200 PRINT !0, 23, "Good. <ENTER> = new values, <CLEAR> = size select, or <E> =
       end.";
21300 ON STOP GOTO "END"
21400 LABEL "Wait"
21500 BW = INKEY$
21600 IF BW = CHR$(013) THEN "Column"
21700 IF BW = "E" OR BW = "e" THEN "END"
21800 IF BW <> CHR$(031) THEN "Wait"
21900 CLS
22000 ERASE H(, V(, M(, I(
22100 GOTO "Length"
22200 LABEL "Partial Total"
22300 'init row and col
22400 H(Y) = 0
22500 V(X) = 0
22600 'calc row and col
22700 FOR J = 1 TO N
22800 H(Y) = H(Y) + M((J - 1) * N + Y)
22900 V(X) = V(X) + M((X - 1) * N + J)
23000 NEXT
23100 GOSUB "MSPOT"
23200 IF H(Y) < -999 OR V(X) < -999 OR H(Y) > 9999 OR V(X) > 9999 THEN K = -1:GO
      TO "Over"
23300 GOSUB "SIH"
23400 GOSUB "SIV"
23500 LABEL "Diagonals"
23600 IF P = 2 AND DI = 2 THEN RETURN
23700 'init diags
23800 LR = 0
23900 LL = 0
24000 'calc diags
24100 FOR F = 1 TO N
24200 LR = LR + M((F - 1) * N + F)
24300 LL = LL + M((N - F) * N + F)
24400 NEXT
24500 IF LR < -999 OR LL < -999 OR LR > 9999 OR LL > 9999 THEN K = -1:GOTO "Over
      "
24600 DI = 2
24700 PRINT !XL / 2, A * N * 2 + 2, USING BU;LR;
24800 PRINT !0, A * N * 2 + 2, USING BU;LL;
24900 RETURN
25000 LABEL "SIH"
25100 PRINT !XL / 2 + 1, Y * A * 2 - A + 1, USING "#####";H(Y);
25200 RETURN
25300 LABEL "MSPOT"
25400 PRINT !X * S / 2 + INT(4 - S / 4) - 2, Y * A * 2 - A + 1, USING BU;M((X - 
      1) * N + Y);
25500 RETURN
25600 LABEL "SIV"
25700 PRINT !X * S / 2 + INT(4 - S / 4) - 2, A * N * 2 + 2, USING BU;V(X);
25800 RETURN
25900 LABEL "USER"
26000 PRINT !0, 22, CHR$(030)
26100 LINE INPUT !0, 23, #4, 95, USING B + "-", "Starting integer ";BS
26200 U = POS(0) + LEN(BS) - 4
26300 L = VAL(BS)
26400 IF L = 0 THEN BS = "0":PRINT !17, 23, BS;:U = POS(0)
26500 LINE INPUT !U, 23, %Q, 3, 95, USING B + "-", ", Increment ";BI
26600 R = VAL(BI)
26700 IF R = 0 THEN "USER"
26800 IF Q = 8 THEN "USER"
26900 IF N * N * R + L - R > 9999 THEN "USER"'max value
27000 Z = (N * N * N * R + N * R + 2 * N * (L - R)) / 2
27100 IF Z < -999 OR Z > 9999 THEN "USER"'total values
27200 G = N / 2
27300 IF G = N / 2 THEN "EVEN"
27400 Y = 1
27500 O = L
27600 FOR T = 1 TO N
27700 FOR W = 1 TO N
27800 X = W + G
27900 IF X > N THEN X = X - N
28000 M((X - 1) * N + Y) = O
28100 PRINT !X * S / 2 + INT(4 - S / 4) - 2, Y * A * 2 - A + 1, USING BU;O;
28200 Y = Y - 1
28300 IF Y = 0 THEN Y = N
28400 O = O + R
28500 NEXT
28600 G = G - 1
28700 IF G < 0 THEN G = G + N
28800 Y = Y + 2
28900 IF Y > N THEN Y = Y - N
29000 NEXT
29100 GOTO "TOTALS"
29200 LABEL "EVEN"
29300 C = INT(N / 4) = N / 4
29400 O = N * N * R + L - R
29500 FOR Y = 1 TO N
29600 FOR X = 1 TO N
29700 E = 0
29800 IF ABS(N - 2 * Y + 1) = N / 2 THEN E = X:X = N - X + 1
29900 M((X - 1) * N + Y) = O
30000 O = O - R
30100 IF E THEN X = E
30200 NEXT
30300 NEXT
30400 GOSUB "MATRIX"
30500 FOR Y = 1 TO N / 2
30600 IF Y AND 1 THEN FOR W = N / 2 TO 1 STEP -1 ELSE FOR W = 1 TO N
30700 IF Y = W OR Y + W - 1 = N THEN "Skip COL"
30800 GOSUB "ROW swap"
30900 GOSUB "ROW swap"
31000 LABEL "Skip COL"
31100 NEXT
31200 NEXT
31300 D = N
31400 IF C THEN D = N / 2
31500 FOR X = 1 TO N / 2
31600 FOR W = 1 TO D
31700 IF X = W OR X + W - 1 = N THEN "Skip ROW"
31800 GOSUB "COL swap"
31900 IF C THEN GOSUB "COL swap"
32000 LABEL "Skip ROW"
32100 NEXT
32200 IF V(X) <> Z THEN GOSUB "Adjust COL"
32300 NEXT
32400 RETURN
32500 LABEL "MATRIX"
32600 FOR X = 1 TO N
32700 FOR Y = 1 TO N
32800 GOSUB "MSPOT"
32900 NEXT
33000 NEXT
33100 LABEL "TOTALS"
33200 ZERO H(, V(
33300 LL = 0
33400 LR = 0
33500 FOR Y = 1 TO N
33600 FOR X = 1 TO N
33700 H(Y) = H(Y) + M((X - 1) * N + Y)
33800 V(X) = V(X) + M((X - 1) * N + Y)
33900 IF Y = N THEN GOSUB "SIV"
34000 NEXT
34100 GOSUB "SIH"
34200 NEXT
34300 GOTO "Diagonals"
34400 LABEL "ROW swap"
34500 X = W
34600 E = M((X - 1) * N + N - Y + 1) - M((X - 1) * N + Y)
34700 IF H(Y) <> Z THEN T = M((X - 1) * N + Y):M((X - 1) * N + Y) = M((X - 1) * 
      N + N - Y + 1):GOSUB "Partial Total":M((X - 1) * N + N - Y + 1) = T:T = Y:
      Y = N - Y + 1:GOSUB "Partial Total":Y = T
34800 W = N - W + 1
34900 RETURN
35000 LABEL "COL swap"
35100 Y = W
35200 E = M((N - X) * N + Y) - M((X - 1) * N + Y)
35300 IF V(X) >= Z THEN IF V(X) + E < Z OR E > 0 THEN "Skip swap"
35400 IF V(X) < Z THEN IF V(X) + E > Z OR E < 0 THEN "Skip swap"
35500 LABEL "DOIT"
35600 T = M((X - 1) * N + Y)
35700 M((X - 1) * N + Y) = M((N - X) * N + Y)
35800 GOSUB "Partial Total"
35900 M((N - X) * N + Y) = T
36000 T = X
36100 X = N - X + 1
36200 GOSUB "Partial Total"
36300 X = T
36400 LABEL "Skip swap"
36500 IF C THEN W = N - W + 1
36600 RETURN
36700 LABEL "Adjust COL"
36800 FOR Y = 1 TO D - 3
36900 IF X = Y OR X + Y - 1 = N THEN "Skip it"
37000 E = M((N - X) * N + Y) - M((X - 1) * N + Y)
37100 F = M((N - X) * N + Y + 1) - M((X - 1) * N + Y + 1)
37200 G = M((N - X) * N + Y + 2) - M((X - 1) * N + Y + 2)
37300 J = M((N - X) * N + Y + 3) - M((X - 1) * N + Y + 3)
37400 IF E + F + V(X) = Z THEN GOSUB "DOIT":GOTO "UP1"
37500 IF E + G + V(X) = Z THEN GOSUB "DOIT":GOTO "UP2"
37600 IF E + J + V(X) = Z THEN GOSUB "DOIT":GOTO "UP3"
37700 IF F + G + V(X) = Z THEN GOSUB "UP1":GOTO "UP1"
37800 IF F + J + V(X) = Z THEN GOSUB "UP1":GOTO "UP2"
37900 IF G + J + V(X) = Z THEN GOSUB "UP2":GOTO "UP1"
38000 IF E + F + G + V(X) = Z THEN GOSUB "DOIT":GOSUB "UP1":GOTO "UP1"
38100 IF E + F + J + V(X) = Z THEN GOSUB "DOIT":GOSUB "UP1":GOTO "UP2"
38200 IF E + G + J + V(X) = Z THEN GOSUB "DOIT":GOSUB "UP2":GOTO "UP1"
38300 IF F + G + J + V(X) = Z THEN GOSUB "UP1":GOSUB "UP1":GOTO "UP1"
38400 LABEL "Skip it"
38500 NEXT
38600 RETURN
38700 LABEL "UP1"
38800 Y = Y + 1
38900 GOTO "DOIT"
39000 LABEL "UP2"
39100 Y = Y + 2
39200 GOTO "DOIT"
39300 LABEL "UP3"
39400 Y = Y + 3
39500 GOTO "DOIT"
39600 LABEL "Fail"
39700 K = I(U) / N - (I(U) / N > INT(I(U) / N))
39800 J = I(U) - N * (K - 1)
39900 W = I(U + 1) / N - (I(U + 1) / N > INT(I(U + 1) / N))
40000 T = I(U + 1) - N * (W - 1)
40100 PRINT !0, 23, "Locations (" RIGHT$(STR$(K), LEN(STR$(K)) - 1)"," RIGHT$(ST
      R$(J), LEN(STR$(J)) - 1)") and (" RIGHT$(STR$(W), LEN(STR$(W)) - 1)"," RIG
      HT$(STR$(T), LEN(STR$(T)) - 1)") are both"M(I(U));
40200 EXIT "Column"
40300 LABEL "END"
40400 ON STOP GOTO 0
40500 CLS
